http://www.mapleprimes.com/questions/137952-How-To-Find-The-Limit

Reference for asymptotic integrals:
- Wong, Asymptotic approximations of integrals (2001), p.55 ff
- Avramidi, Notes on Asymptotic Expansions (2001), http://www.nmt.edu/%7Eiavramid/notes/asexp.pdf
- Malham, An introduction to asymptotic analysis, http://www.ma.hw.ac.uk/~simonm/ae.pdf

Reference for asymptotic 2F1:
- Temme, Large Parameter Cases of the Gauss Hypergeometric Function (2002), http://arxiv.org/abs/math/0205065v1

```
> restart; interface(version);
  Digits:=15;
  with(IntegrationTools):
```

Classic Worksheet Interface, Maple 16.00, Windows, Mar 3 2012, Build ID 732982

$$Digits := 15$$

```
> `Assertion:`;
  n*Int((1-(3/26)*x-(37/49)*x^2)^n, x = 0 .. 1):
  Limit(%, n=infinity) = `8` + 2/3;
```

Assertion:

$$\lim_{n \to \infty} n \int_0^1 \left(1 - \frac{3}{26} x - \frac{37}{49} x^2\right)^n dx = 8 + \frac{2}{3}$$

```
> # integrand
  (1-(3/26)*x-(37/49)*x^2); factor(%); #%^1;
  plot(%, x=0..1);
```

$$1 - \frac{3}{26} x - \frac{37}{49} x^2$$

$$-\frac{(74 x + 91)(13 x - 14)}{1274}$$



```
> # so do a change of variables:
  t=(1-(3/26)*x-(37/49)*x^2); [solve(%, x)];
  simplify(subs(t=1, %)); # for choosing the correct one
```

$$t = 1 - \frac{3}{26} x - \frac{37}{49} x^2$$

$$\left[-\frac{147}{1924} + \frac{7\sqrt{100489 - 100048\, t}}{1924}, -\frac{147}{1924} - \frac{7\sqrt{100489 - 100048\, t}}{1924}\right]$$

$$\left[0, \frac{-147}{962}\right]$$

```
> n*Int((1-(3/26)*x-(37/49)*x^2)^n, x = 0 .. 1);
  ``=Change(%, x= -147/1924+7/1924*(100489-100048*t)^(1/2), t);
  value(%): expand(%):
  collect(%, hypergeom);
  #A:=op(1,Sol);
  #B:=op(2,Sol);
  #'tmp=A+B'; is(%);
  Sol:=rhs(%):
```

$$n \int_0^1 \left(1 - \frac{3}{26}x - \frac{37}{49}x^2\right)^n dx$$

$$= 182\, n \int_{\frac{165}{1274}}^1 \frac{t^n}{\sqrt{100489 - 100048\, t}}\, dt$$

$$= -\frac{165}{2219} \frac{n \,\mathrm{hypergeom}\left(\left[\frac{1}{2}, n+1\right], [2+n], \frac{634920}{4923961}\right) 165^n}{(n+1)\, 1274^n} + \frac{182}{317} \frac{n \,\mathrm{hypergeom}\left(\left[\frac{1}{2}, n+1\right], [2+n], \frac{100048}{100489}\right)}{n+1}$$

Temme, p. 3/4:

$$\begin{aligned} {}_2F_1\left({a, \beta+\lambda \atop \gamma+\lambda}; z\right) &= (1-z)^{-a}\,{}_2F_1\left({a, \gamma-\beta \atop \gamma+\lambda}; \frac{z}{z-1}\right) \\ &= (1-z)^{-a}\left[1 + \frac{a(\gamma-\beta)}{\gamma+\lambda}\frac{z}{z-1} + \frac{(a)(a+1)(\gamma-\beta)(\gamma-\beta+1)}{(\gamma+\lambda)(\gamma+\lambda+1)\,2!}\left(\frac{z}{z-1}\right)^2 \ldots\right] \qquad (2.5) \\ &= (1-z)^{-a}\left[1 + \frac{a(\gamma-\beta)}{\gamma+\lambda}\frac{z}{z-1} + \mathcal{O}\left(\lambda^{-2}\right)\right], \end{aligned}$$

"It is an asymptotic expansion for lambda large, and all fixed z, z <> 1."

```
> (1-z)^(-a)*(1 + a*(gamma - beta)/(gamma+lambda)*z/(z-1) );
  ``=subs(lambda=n, gamma=2, beta=1, %);
  #subs(a=1/2, %);
  rhs(%):
  Limit(%, n=infinity): '%'=value(%);
  ``=subs(a=1/2, rhs(%));
```

$$(1-z)^{(-a)}\left(1 + \frac{a(\gamma-\beta)z}{(\gamma+\lambda)(z-1)}\right)$$

$$= (1-z)^{(-a)}\left(1 + \frac{a z}{(2+n)(z-1)}\right)$$

$$\lim_{n \to \infty} (1-z)^{(-a)}\left(1 + \frac{a z}{(2+n)(z-1)}\right) = \frac{1}{(1-z)^a}$$

$$= \frac{1}{\sqrt{1-z}}$$

```
> # thus we have the rule
  hypergeom([1/2, n+1],[2+n],z):
  Limit(%, n=infinity) = evalindets(%, 'specfunc( anything, hypergeom )', f -> 1/sqrt(1 - op(3,f)));
```

$$\lim_{n \to \infty} \mathrm{hypergeom}\left(\left[\frac{1}{2}, n+1\right], [2+n], z\right) = \frac{1}{\sqrt{1-z}}$$

```
>
> Limit(n*Int((1-3/26*x-37/49*x^2)^n,x = 0 .. 1),n = infinity);
  ``=Limit(Sol, n=infinity);
  evalindets(%, 'specfunc( anything, hypergeom )', f -> 1/sqrt(1 - op(3,f)));
  value(%);
```

$$\lim_{n \to \infty} n \int_0^1 \left(1 - \frac{3}{26}x - \frac{37}{49}x^2\right)^n dx$$

$$= \lim_{n \to \infty} \left(-\frac{165}{2219} \frac{n \,\mathrm{hypergeom}\left(\left[\frac{1}{2}, n+1\right], [2+n], \frac{634920}{4923961}\right) 165^n}{(n+1)\, 1274^n} + \frac{182}{317} \frac{n \,\mathrm{hypergeom}\left(\left[\frac{1}{2}, n+1\right], [2+n], \frac{100048}{100489}\right)}{n+1} \right)$$

$$= \lim_{n \to \infty} \left(-\frac{165\, n\, 165^n}{2071\, (n+1)\, 1274^n} + \frac{26\, n}{3\, (n+1)} \right)$$

$$= \frac{26}{3}$$

testing

```
> n*Int((1-(3/26)*x-(37/49)*x^2)^n, x = 0 .. 1):
  combine(%); subs(n=100, %);
  plot(op(%));
```

$$\int_0^1 n \left(1 - \frac{3}{26} x - \frac{37}{49} x^2\right)^n dx$$

$$\int_0^1 100 \left(1 - \frac{3}{26} x - \frac{37}{49} x^2\right)^{100} dx$$



That integral is roughly = 8. Estimate, where one can cut off

```
> eps:='eps':
  eps=n*(X)^n; isolate(%, X);
  subs(X=1-3/26*x-37/49*x^2, %);
  [solve(%, x)]; #simplify(%, symbolic);
  Upper:=%[1];
```

$$eps = n\, X^n$$

$$X = \left(\frac{eps}{n}\right)^{\left(\frac{1}{n}\right)}$$

$$1 - \frac{3}{26} x - \frac{37}{49} x^2 = \left(\frac{eps}{n}\right)^{\left(\frac{1}{n}\right)}$$

$$\left[-\frac{147}{1924}+\frac{7\sqrt{100489-100048\left(\frac{eps}{n}\right)^{\left(\frac{1}{n}\right)}}}{1924},\ -\frac{147}{1924}-\frac{7\sqrt{100489-100048\left(\frac{eps}{n}\right)^{\left(\frac{1}{n}\right)}}}{1924}\right]$$

$$Upper := -\frac{147}{1924}+\frac{7\sqrt{100489-100048\left(\frac{eps}{n}\right)^{\left(\frac{1}{n}\right)}}}{1924}$$

```
> oldDigits:=Digits:
  Digits:=3*Digits:

  kTst:=16;
  nTst:='10^kTst'; nTst:=evalf(nTst);
  eps:= 1e-12;
  ``;
  #eps:='eps';
  'Int(n*(1-3/26*x-37/49*x^2)^n,x = 0 .. 1)';
  ``='Int(n*(1-3/26*x-37/49*x^2)^n,x = 0 .. Upper)';
  subs(n=evalf(nTst), %);

  rhs(%):
  Int(op(%), epsilon=eps):
  evalf(%):
  ``=evalf[oldDigits](%);

  Digits:=oldDigits:
```

$$kTst := 16$$

$$nTst := 10^{kTst}$$

$$nTst := 0.100000000000000\ 10^{17}$$

$$eps := 0.100000000000000\ 10^{-11}$$

$$\int_0^1 n\left(1-\frac{3}{26}x-\frac{37}{49}x^2\right)^n dx$$

$$= \int_0^{Upper} n\left(1-\frac{3}{26}x-\frac{37}{49}x^2\right)^n dx$$

$$= \int_0^{0.558760649233016\ 10^{-13}} 0.100000000000000\ 10^{17}\left(1-\frac{3}{26}x-\frac{37}{49}x^2\right)^{0.100000000000000\ 10^{17}} dx$$

$$= 8.666666666666570$$

```
> 
> 
```